


```

; the TVRD() function uses an X routine that has problems if the
; window is obscured or iconified. This routine uses the DEVICE, COPY=
; command to first copy the window contents to a new window pixmap,
; and then reads from this pixmap into the array.
;
; Modifications
;
; 6-11-97 DSF Check validity of draw widget.

```

```

FUNCTION safe_tvrd, draw_widget, xsize, ysize

```

```

on_error, 2

```

```

if (widget_info(draw_widget, /valid_id) eq 0) then begin
  return, -1
endif else if (widget_info(draw_widget, /name) ne 'DRAW') then begin
  return, -1
endif else begin
  old_window = !d.window
  window, xsize=xsize, ysize=ysize, /free, /pixmap ; Create new window

  widget_control, draw_widget, get_value=window
  device, copy=[0,0, xsize,ysize, 0,0, window] ; Copy into new window

  image = tvrd() ; Read into array
  wdelete, !d.window
  if (old_window ne -1) then wset, old_window
  return, image
endelse

```

```

END

```

```

SAFE_TVIRD

```

This function replaces IDL's TVRD() function for reading the contents of a window into an array. The TVRD() function returns unexpected results if the window is scrollable or is obscured onscreen. This function uses IDL's DEVICE, COPY=[] function to more safely read the window contents.

Calling Sequence

```

Array = SAFE_TVIRD(Draw_widget, Xsize, Ysize)

```

Arguments

```

Draw_widget

```

The widget id of the draw widget which you will be reading into the array. Note that this is NOT the window id!

Xsize, Ysize

The dimensions of the window in the draw widget.

Outputs

Array

Returns the image read from the draw widget. Returns -1 if the Draw_widget is not a valid draw widget ID.

File Attachments

- 1) [safe_tvrd.pro](#), downloaded 108 times
 - 2) [safe_tvrd.doc](#), downloaded 140 times
-